

Paper 116-2011

Creating a Table of Contents for MS Word using AutoFormat and Cascading Style Sheets

Fareeza Khurshed, BC Cancer Agency, Vancouver, British Columbia

ABSTRACT

This paper demonstrates how to create a table of contents (TOC) with SAS® TITLE statements, cascading style sheets (CSS), and Microsoft Word. The CSS mimic Microsoft Word styles allowing the Microsoft Word AutoFormat to classify them as appropriate Microsoft Word styles. Once the styles are implemented, built-in Word functionality can be used to create a TOC containing the appropriate page references and hyperlinks to the relevant sections.

INTRODUCTION

This paper demonstrates how to use CSS and AutoFormat in Word to create a TOC. It presents a brief overview of style sheets, how to find the SAS style sheets and the code to implement Word heading formats in a style sheet. A summary of the various methods of creating a TOC is also presented. This method proposed in this paper can incorporate different levels of headings in a TOC and will automatically create heading for BY groups, if desired. All methods and code in the paper have been tested using SAS 9.2, Windows XP Professional SP3, and Microsoft Office 2003.

OVERVIEW OF CASCADING STYLE SHEETS

What is a style sheet? A Cascading Style Sheet is a set of instructions that control the formatting of content and objects in a document. The style allows the separation of the document content and document styling, such as layout, fonts and colours. CSS is primarily used in web design with thousands of templates available online. SAS provides a library of CSS templates on install found where SAS is installed. Depending on the install settings the locations of the styles sheets can vary, but on my PC they are located at:

...\SAS\SharedFiles\BIClientStyles\...

If you know the template you are interested in, you can use SAS code to generate a copy of the style sheet. The following code will create a copy of the style sheet in the C:\temp directory.

```
ods html path='C:\temp' (url=none)
  file='makecss.html'
  style=journal
  stylesheet='copy_of_journal.css';

proc print data=sashelp.class(obs=1);
run;

ods html close;
```

SAS Enterprise Guide® has a style editor that shows the full catalogue of style sheets available to SAS users by default and available to be modified. The benefit of using a style sheet is enormous as the code can be entirely maintained outside of SAS and updated whenever required. The CSS code can also be designed to match a company's intranet or external website, making it easier to produce documents that match company design. For a more detailed introduction into CSS and SAS see Cynthia Zender's paper. This paper will demonstrate how to modify the style options in a CSS template to match the Word styles and automatically create a TOC.

CREATING A STYLE SHEET

To create your own style sheet, you need to know the specific class names that SAS uses for each document element, such as **.tableheader**, **.title**, or **.byline**. The SAS style sheet 'meadow' has 164 different classes. Given the number of classes available for modification in a style sheet, it is much easier to take a template that is close to what you want and modify it. The Style Editor within SAS Enterprise Guide provides a GUI for a user to change settings as well. To implement this method of creating a TOC you will have to create your own style sheet.

The goal is to have SAS TITLE statements resolve to MS Word Header 1 styles, SAS TITLE2 statements resolve to MS Word Header 2 styles and so forth. MS Word does have an AutoFormat feature that reformats any style that is similar to a built in style to that style. By formatting the SAS title, by line and text into styles that are similar to MS Word styles the AutoFormat features can be used to create an automated TOC.

These are examples of the heading styles used in MS Word:

Heading 1 Style: Arial, 16pt, bold, left aligned

Heading 2 Style: Arial, 14pt, bold, italicized, left aligned

Heading 3 Style: Arial, 13pt, bold, left aligned

This is the CSS implementation of the MS Word header styles and implementation for BY group processing and TEXT statements.

```

/*Title formatting for Heading 1*/
.systemtitle{
  font-family: Arial, Helvetica, sans-serif;
  font-size:16pt;
  font-weight: bold;
  color: black;
  text-align:left;
}

/*Title2 formatting for Heading 2*/
.systemtitle2{
  font-family: Arial, Helvetica, sans-serif;
  font-size: 14pt;
  font-weight: bold;
  font-style: italic;
  color: black;
  text-align:left;
}

/*Title3 formatting for Heading 3*/
.systemtitle3{
  font-family: Arial, Helvetica, sans-serif;
  font-size:13pt;
  font-weight: bold;
  color: black;
  text-align:left;
}

/*Formats for ods rtf text= statements*/
.usertext{
  font-family: Arial, Helvetica, sans-serif;
  font-size:12pt;
  font-weight: bold;
  border:none;
  color: black;
  text-align:center;
}

/*Formatting for BY Group labels*/
.byline{
  font-family: Arial, Helvetica, sans-serif;
  font-size:11pt;
  font-weight: bold;
  color: black;
  text-align:center;}

```

To create your own style sheet make a copy of the CSS template you would like to use in your report, never make changes to the originals. Open the file in a text editor or SAS and replace the **.systemtitle** classes with the ones above to create the Word Header styles. Additionally, you can modify the **.usertext** and **.byline** classes if you want to include TEXT= and BY groups in your TOC levels.

METHODS TO CREATE TABLE OF CONTENTS

There are currently several methods in SAS to create TOC's for RTF files, which are very clearly detailed in Electra Small's paper, *Creating Table of Contents in RTF Documents*. All methods currently require opening the RTF document and inserting or updating a TOC table. Code to generate the TOC in the figures can be found in Appendix 1.

Using the CONTENTS=YES and TOC_DATA options in the ODS RTF statement and ODS PROCLABEL= statements in the code are used to create the TOC entries. This results in a TOC that has extra information generated by SAS and an excess of labeling. Additionally, because the ODS PROCLABEL= statements are not displayed in the RTF output, if you do want the headings in the output you need to specify TITLE statements as well. The CONTENTS= option within each procedure can help to reduce some of the redundancy but that means extra coding in each procedure and sometimes even modifying the template for procedure, such as one way tables for Proc Freq.

Car Analysis	2
Make=Ford	2
Table Type	2
One-Way Frequencies	2
Make=Honda	3
Table Type	3
One-Way Frequencies	3
Make=Hyundai	4
Table Type	4
One-Way Frequencies	4
Average Mileage	5
Summary statistics	5

Figure 1 Table of contents using CONTENTS=yes and TOC_DATA option

Error! No table of contents entries found. A second method involves using WORDSTYLE option, experimental in SAS 9.1, and commands imbedded in TITLE, TEXT or PREPAGE statements. This method requires a user to type a string of RTF syntax at the beginning of each statement. It does allow multiple levels of TOC's but is still experimental and not documented. It does not handle titles automatically created when using BY groups.

The third and most flexible method involves passing RTF syntax through SAS. This method inserting the RTF commands into the document using RTF TEXT=options. For complex documents that need repeating and where a TOC is always required, this is probably the best method overall, however this does require more knowledge of the RTF syntax than other methods.

Car Analysis	1
By Model	1
Average Mileage	3

Figure 2 Table of contents using the WORDSTYLE option

The method proposed in this paper to create a TOC requires uses the CSSSTYLE= and BODYTITLE options.

```
ods rtf file='C:\temp\temp.rtf'
  CSSSTYLE="C:\...\stylesheet_toc.css"
  BODYTITLE
  Startpage=never;

  title 'Testing Level1';

  ***SAS Program Code*****;

  title2 'Testing Level2';

  ****SAS Program Code*****;

ods rtf close;
```

Assuming the SAS program runs without errors the RTF file is created. Opening the file in Word, select **Format>AutoFormat**. Under **Options** check only **Built-In Heading Styles**. Click **OK**. To insert a TOC click **Insert>Reference>Index and Tables**. A TOC will appear in the document and you're done!

Car Analysis	1
Make=Ford.....	1
Make=Honda.....	1
Make=Hyundai.....	1
Average Mileage	1

Figure 3 Table of content using CSSSTYLE= and BODYTITLE options

COMMENTS

Some SAS procedures output their own title which may not be one of the styles identified so the style sheet would have to be modified. To determine which class would need modifying you can output the file as html, identify the class used in the HTML code and then modify it in the style sheet. If SAS generates titles that appear multiple times in the document then those titles will be repeated in this TOC as well. A method to avoid this is to use the TEXT= method for Header 1 rather than TITLE statements and the style sheet for this method would be different than the one included in this paper.

CONCLUSION

All TOC methods require some manual procedure after opening the file with MS Word or another text editor. The method proposed here consists of 3 steps: creation of a style sheet with the definition of the header styles, inclusion of the relevant title statements in each of the procedures used and the creation of TOC in Word. As the style sheet only needs to be created once and title statements are already familiar to users this method may be useful for beginners and those familiar with MS Word. This method can also incorporate BY groups used in the TOC.

REFERENCES

- Zender, Cynthia. 2009. "CSSSTYLE: Stylish Output with ODS and SAS 9.2". Proceedings of the SAS Global Forum 2009 Conference. Cary, NC. SAS Institute Inc. 014-2009
- Small, Electra. 2007. "Creating Table of Contents in RTF Documents". Proceedings of the SAS Global Forum 2007 Conference. Cary, NC. SAS Institute Inc. 097-2007.

ACKNOWLEDGMENTS

Thanks to Colleen McGahan and Ryan Woods at the BC Cancer Agency for assistance in developing and editing this paper.

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Fareeza Khurshed
BC Cancer Agency
801 – 686 W Broadway
Vancouver, British Columbia V5Z 1G1
fkhurshed@bccancer.bc.ca

APPENDIX 1

```

*TOC VIA CONTENTS=YES Option;
ods rtf file='C:\temp\test_method1.rtf' contents=YES toc_data;
ods noproctitle;

proc sort data=sashelp.cars out=cars;
  by make;
run;

ODS PROCLABEL='Car Analysis';
title 'Car Analysis - Model';
proc freq data=cars (where= (make in ("Honda", "Hyundai", "Ford")));
  by make;
  tables type/contents='';
run;

ODS PROCLABEL= 'Average Mileage';
title 'Average Mileage';
proc means data=sashelp.cars maxdec=2;
  var mpg_city mpg_highway;
run;

ods rtf close;
*Open Word File and update TOC fields;

title;
*****
*****;
*TOC VIA wordstyle Option;
ods rtf file='C:\temp\test_method2.rtf' wordstyle="{\s1 Heading 1;}{\s2 Heading
2;}";
bodytitle;
ods noproctitle;

ods escapechar='\';

proc sort data=sashelp.cars out=cars;
  by make;
run;

ods rtf text='\R/RTF"\s1 " Car Analysis';
ods rtf text='\R/RTF"\s2 " By Model';
proc freq data=cars (where= (make in ("Honda", "Hyundai", "Ford")));
  by make;
  tables type;
run;

ods rtf text='\R/RTF"\s1 " Average Mileage';
proc means data=sashelp.cars maxdec=2;
  var mpg_city mpg_highway;
run;

ods rtf close;

```

```
*****
*****;
*TOC VIA CSSSTYLE= Option;
options orientation=portrait nodate nonumber;
ods rtf file="c:\temp\TOC.rtf"
    cssstyle="H:\PPO\Fareeza\SAS\stylesheet_test3.css"
    bodytitle
    startpage=never
;

ods noproctitle;

proc sort data=sashelp.cars out=cars;
    by make;
run;

title 'Car Analysis';
proc freq data=cars (where= (make in ("Honda", "Hyundai", "Ford")));
    by make;
    tables type;
run;

title 'Average Mileage';
proc means data=sashelp.cars maxdec=2;
    var mpg_city mpg_highway;
run;

ods rtf close;
```

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. © indicates USA registration.

Other brand and product names are trademarks of their respective companies.